NAG Fortran Library Routine Document

F07CAF (DGTSV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CAF (DGTSV) computes the solution to a real system of linear equations

$$AX = B,$$

where A is an n by n tridiagonal matrix and X and B are n by r matrices.

2 Specification

```
SUBROUTINE F07CAF (N, NRHS, DL, D, DU, B, LDB, INFO)
INTEGER            N, NRHS, LDB, INFO
double precision   DL(*), D(*), DU(*), B(LDB,*)
```

The routine may be called by its LAPACK name ***dgtsv***.

3 Description

Gaussian elimination with partial pivoting and row interchanges is used to solve the equations $AX = B$. The matrix A is factorized as $A = PLU$, where P is a permutation matrix, L is unit lower triangular with at most one non-zero sub-diagonal element per column, and U is an upper triangular band matrix, with two super-diagonals.

Note that the equations $A^T X = B$ may be solved by interchanging the order of the arguments DU and DL.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: $\mathrm{NRHS} \geq 0$.

3: DL($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array DL must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n-1)$ sub-diagonal elements of the matrix A.

On exit: if $\mathrm{INFO} \geq 0$, DL is overwritten by the $(n-2)$ elements of the second super-diagonal of the upper triangular matrix U from the LU factorization of A in $\mathrm{DL}(1), \mathrm{DL}(2), \ldots, \mathrm{DL}(n-2)$.

4: D(∗) – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the matrix A.

On exit: if INFO ≥ 0, D is overwritten by the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

5: DU(∗) – ***double precision*** array *Input/Output*

Note: the dimension of the array DU must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n - 1)$ super-diagonal elements of the matrix A.

On exit: if INFO ≥ 0, DU is overwritten by the $(n - 1)$ elements of the first super-diagonal of U.

6: B(LDB,∗) – ***double precision*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$. To solve the equations $Ax = b$, where b is a single right-hand side, B may be supplied as a one-dimensional array with length $\mathrm{LDB} = \max(1, \mathrm{N})$.

On entry: the n by r matrix of right-hand sides B.

On exit: if INFO $= 0$ or INFO $= \mathrm{N} + 1$, the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07CAF (DGTSV) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

8: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, u_{ii} is exactly zero. The factorization has been completed, but the factor U is exactly singular, so the solution could not be computed.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

Alternatives to F07CAF (DGTSV), which return condition and error estimates are F04BCF and F07CBF (DGTSVX).

8 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ is proportional to nr.

The complex analogue of this routine is F07CNF (ZGTSV).

9 Example

To solve the equations

$$Ax = b,$$

where A is the tridiagonal matrix

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix}$$

and

$$b = \begin{pmatrix} 2.7 \\ -0.5 \\ 2.6 \\ 0.6 \\ 2.7 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07CAF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*     .. Local Scalars ..
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION B(NMAX), D(NMAX), DL(NMAX-1), DU(NMAX-1)
*     .. External Subroutines ..
      EXTERNAL         DGTSV
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07CAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the tridiagonal matrix A and the right hand side B from
*        data file
*
         READ (NIN,*) (DU(I),I=1,N-1)
         READ (NIN,*) (D(I),I=1,N)
```

```
         READ (NIN,*) (DL(I),I=1,N-1)
         READ (NIN,*) (B(I),I=1,N)
*
*        Solve the equations Ax = b for x
*
         CALL DGTSV(N,1,DL,D,DU,B,N,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Solution'
            WRITE (NOUT,99999) (B(I),I=1,N)
*
         ELSE
            WRITE (NOUT,99998) 'The (', INFO, ',', INFO, ')',
     +        ' element of the factor U is zero'
         END IF
       ELSE
          WRITE (NOUT,*) 'NMAX too small'
       END IF
       STOP
*
99999 FORMAT ((1X,7F11.4))
99998 FORMAT (1X,A,I3,A,I3,A,A)
      END
```

9.2 Program Data

```
F07CAF Example Program Data
  5                              :Value of N
       2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0         :End of matrix A
  2.7  -0.5   2.6   0.6   2.7  :End of vector B
```

9.3 Program Results

```
 F07CAF Example Program Results

 Solution
     -4.0000     7.0000     3.0000    -4.0000    -3.0000
```